
157  Interviews 1: Print

Fractal.Invaders, Substrate  (Interview with Jared Tarbell)

 Creator  Jared Tarbell
 Year  2004
 Medium  Software, Prints
 Software  Flash, Processing
 URL  www.complexification.net

 What are Fractal.Invaders and Substrate?
 Fractal.Invaders and Substrate are unique programs that both generate space-filling 
patterns on a two-dimensional surface. Each uses simplified algorithmic processes to render 
a more complex whole.  
 Fractal.Invaders begins with a rectangular region and recursively fills it with little 
“invader” objects. Each invader is a combination of black squares arranged in a 5 * 5 grid 
generated at random during runtime. The only rule of construction requires that the left side 
of the invader be a mirror copy of the right side. This keeps them laterally symmetric, which 
endows them with a special attractiveness to the human eye.
 There are a total of 32,768 (215) possible invaders. The magnitude of 15 comes from the 
product of 3 columns and 5 rows (the last 2 columns of the grid are ignored since they are the 
same as the first 2). The 2 comes from the fact that each space in the grid can be either black 
or white.
 A small bit of interactivity allows each invader to be clicked. Clicking an invader destroys it, 
although the empty space left behind is quickly filled with smaller invaders. In this way, the user 
is ultimately doomed.
 Substrate begins similarly with an empty rectangular region. It has been compared 
to crystal formation and the emergent patterns of urban landscapes. A single line (known 
internally as a “crack” since the algorithm was inspired by sunbaked mud cracks) begins 
drawing itself from some random point in some random direction. The line continues to draw 
itself until it either (a) hits the edge of the screen or (b) hits another line, at which point it stops 
and two more lines begin. The one simple rule used in the creation of new lines is that they 
begin at tangents to existing lines. This process is repeated until there are too many lines to keep 
track of or the program is stopped. 
 Before writing the program, I only had a vague idea of what it might look like. It wasn’t 
until the first couple of bug-free executions that I realized something incredible was happening. 
The resulting form was much more complex than the originating algorithm. This particular 
quality of software is what keeps me interested.
 Interesting effects can be created by introducing small variations in the way the first couple 
of lines are drawn. One of my favorite initial conditions is the creation of three lines, each in 
its own localized space with a direction that varies from the others by about 30 degrees. After 
growing for a short time into coherent lattices, they eventually crash into each other, creating 
an affluence of odd shapes and unexpected mazes.
 The watercolor quality of the rendering is achieved by placing large numbers of mostly 
transparent pixels perpendicular to each line’s growth. The trick is to deposit precisely the same Su

158  Interviews 1: Print

number of pixels regardless of the length of the area being filled. This produces an interesting 
density modulation across an even mass of pixels.
 Why did you create this software? 
 For me, one of the most enjoyable subjects in computer science is combination. I ask myself 
a question like, “Given some rules and a few simple objects, how many possible ways can they 
be combined?” Seldom can I answer this using thought alone, mainly because the complexity 
of even just a few elements is outside the realm of my imagination. Instead, I write computer 
programs to solve it for me. Fractal.Invaders is definitely one of these questions, and is answered 
completely with the rendering of every single invader. Substrate asks a similar question but with 
results that, although beautiful, are a little less complete. 
 What software tools were used? 
 For Fractal.Invaders, I used a combination of Flash and custom software to create and 
capture the invaders, respectively. In Flash, all work was done using ActionScript. A single 
symbolic element (black square) exists in the library. Code takes this square and duplicates 
it hundreds of thousands of times. The entire generative process takes about five minutes 
to complete, depending on the size of the region to be filled and the speed of the execution. 
Capturing a high-resolution image of the result is accomplished with a program that scales the 
Shockwave Flash (SWF) file very large and saves the screen image out to a file.
 Substrate was created entirely in Processing. Processing was particularly well suited for this 
as it excels at drawing, especially when dropping millions of deep-color pixels. Processing can 
also save out extremely large graphic images in an automated fashion. Oftentimes I will run a 
Processing project overnight. In the morning I awake to a vast collection of unique images, the 
best of which are archived as print editions.
  Why did you use these tools?
 I use Flash because I am comfortable working within it. I use Processing because it enables 
me to do things Flash simply cannot. Both environments allow me to take a program from 
concept to completion in a number of hours. Complex visual logic can be built up without the 
bulky overhead required in more traditional graphic programming languages. 
 Flash excels at rendering very high resolution images nicely, displaying native vector 
objects with a high degree of precision and full antialiasing. Processing gives me the 
computational speed to increase the number of objects in the system by a magnitude of 20 
or more. Both programs allow me to produce work that is capable of being viewed by a large 
number of people worldwide.
 Why do you choose to work with software?
 With software, anything that can be imagined can be built. Software has a mysterious, 
undefined border. Programming is truly a process of creating something from nothing. I enjoy 
most John Maeda’s perspective: “While engaged in the deepest trance of coding, all one needs to 
wish for is any kind of numerical or symbolic resource, and in a flash of lightning it is suddenly 
there, at your disposal.”

Fractal.Invaders, 2004. Image courtesy of the artist. 

205  

Input 1:  Mouse I
This unit introduces mouse input as a way to control the position and attributes of shapes 
on screen. It also explains how to change the cursor icon.

Syntax introduced:
mouseX, mouseY, pmouseX, pmouseY, mousePressed, mouseButton

cursor(), noCursor()

The screen forms a bridge between our bodies and the realm of circuits and electricity 
inside computers. We control elements on screen through a variety of devices such as 
touch pads, trackballs, and joysticks, but—aside from the keyboard—the most common 
input device is the mouse. The computer mouse dates back to the late 1960s when 
Douglas Engelbart presented the device as an element of the oN-Line System (NLS), one 
of the fi rst computer systems with a video display. The mouse concept was further 
developed at the Xerox Palo Alto Research Center (PARC), but its introduction with the 
Apple Macintosh in 1984 was the catalyst for its current ubiquity. The design of the 
mouse has gone through many revisions in the last thirty years, but its function has 
remained the same. In Engelbart’s original patent application in 1970 he referred to the 
mouse as an “X-Y position indicator,” and this still accurately, but dryly, defi nes its 
contemporary use. 
 The physical mouse object is used to control the position of the cursor on screen and 
to select interface elements. The cursor position is read by computer programs as two 
numbers, the x-coordinate and the y-coordinate. These numbers can be used to control 
attributes of elements on screen. If these coordinates are collected and analyzed, they 
can be used to extract higher-level information such as the speed and direction of the 
mouse. This data can in turn be used for gesture and pattern recognition.

Mouse data 

The Processing variables mouseX  and mouseY (note the capital X and Y) store the 
x-coordinate and y-coordinate of the cursor relative to the origin in the upper-left corner 
of the display window. To see the actual values produced while moving the mouse, run 
this program to print the values to the console:

void draw() {

  frameRate(12);

  println(mouseX + " : " + mouseY);

}

23-01

206  Input 1: Mouse I

When a program starts, mouseX and mouseY values are 0. If the cursor moves into the 
display window, the values are set to the current position of the cursor. If the cursor is
at the left, the mouseX value is 0 and the value increases as the cursor moves to the right. 
If the cursor is at the top, the mouseY value is 0 and the value increases as the cursor 
moves down. If mouseX and mouseY are used in programs without a draw() or if 
noLoop() is run in setup(), the values will always be 0.
 The mouse position is most commonly used to control the location of visual 
elements on screen. More interesting relations are created when the visual elements 
relate differently to the mouse values, rather than simply mimicking the current 
position. Adding and subtracting values from the mouse position creates relationships 
that remain constant, while multiplying and dividing these values creates changing 
visual relationships between the mouse position and the elements on the screen. To 
invert the value of the mouse, simply subtract the mouseX value from the width of the 
window and subtract the mouseY value from the height of the screen. 

  // Circle follows the cursor (the cursor position is

  // implied by the crosshairs around the illustration)

  void setup() {

    size(100, 100);

    smooth();

    noStroke();

  }

  void draw() {

    background(126);

    ellipse(mouseX, mouseY, 33, 33);

  }

  // Add and subtract to create offsets

  void setup() {

    size(100, 100);

    smooth();

    noStroke();

  }

  void draw() {

    background(126);

    ellipse(mouseX, 16, 33, 33);     // Top circle

    ellipse(mouseX+20, 50, 33, 33);  // Middle circle

    ellipse(mouseX-20, 84, 33, 33);  // Bottom circle

  }

23-02

207  Input 1: Mouse I

  // Multiply and divide to creates scaling offsets

  void setup() {

    size(100, 100);

    smooth();

    noStroke();

  }

  void draw() {

    background(126);

    ellipse(mouseX, 16, 33, 33);    // Top circle

    ellipse(mouseX/2, 50, 33, 33);  // Middle circle

    ellipse(mouseX*2, 84, 33, 33);  // Bottom circle

  }

  // Invert cursor position to create a secondary response

  void setup() { 

    size(100, 100); 

    noStroke(); 

    smooth();

  } 

 

  void draw() {   

    float x = mouseX;

    float y = mouseY;

    float ix = width - mouseX;   // Inverse X

    float iy = mouseY - height;  // Inverse Y

    background(126); 

    fill(255, 150); 

    ellipse(x, height/2, y, y); 

    fill(0, 159); 

    ellipse(ix, height/2, iy, iy); 

  } 

23-04

208  Input 1: Mouse I

  // Exponential functions can create nonlinear relations  

  // between the mouse and shapes affected by the mouse 

  void setup() {

    size(100, 100);

    smooth();

    noStroke();

  }

  void draw() {

    background(126);

    float normX = mouseX / float(width);

    ellipse(mouseX, 16, 33, 33);                 // Top

    ellipse(pow(normX, 4) * width, 50, 33, 33);  // Middle

    ellipse(pow(normX, 8) * width, 84, 33, 33);  // Bottom

  }

The Processing variables pmouseX and pmouseY store the mouse values from the 
previous frame. If the mouse does not move, the values will be the same, but if the 
mouse is moving quickly there can be large differences between the values. To see the 
difference, run the following program and alternate moving the mouse slowly and 
quickly. Watch the values print to the console.

void draw() {

  frameRate(12);

  println(pmouseX - mouseX);

}

Drawing a line from the previous mouse position to the current position shows the 
changing position in one frame, revealing the speed and direction of the mouse. When the 
mouse is not moving, a point is drawn, but quick mouse movements create long lines.

  // Draw a line between the current and previous positions

  void setup() {

    size(100, 100);

    strokeWeight(8);

    smooth();

  }

  void draw() {

    background(204);

    line(mouseX, mouseY, pmouseX, pmouseY);

  }

23-06

209  Input 1: Mouse I

The mouseX and mouseY values can control translation, rotation, and scale by using 
them as parameters in the transformation functions. You can move a circle around the 
screen by changing the parameters to translate() rather than by changing the x and 
y parameters of ellipse().

  // Use translate() to move a shape

  void setup() {

    size(100, 100);

    smooth();

    noStroke();

  }

  void draw() {

    background(126);

    translate(mouseX, mouseY);

    ellipse(0, 0, 33, 33);

  }

Before using mouseX and mouseY as parameters to transformation functions, it’s 
important to think fi rst about how they relate to the expected parameters. For example, 
the rotate() function expects its parameters in units of radians (p. 117). To make a 
shape rotate 360 degrees as the cursor moves from the left edge to the right edge of the 
window, the values of mouseX must be converted to values from 0.0 to 2π. In the 
following example, the map() function is used to make this conversion. The resulting 
value is used as the parameter to rotate() to turn the line as the mouse moves back 
and forth between the left and right edge of the display window.

  // Use rotate() to move a shape

  void setup() {

    size(100, 100);

    strokeWeight(8);

    smooth();

  }

  void draw() {

    background(204);

    float angle = map(mouseX, 0, width, 0, TWO_PI);

    translate(50, 50);

    rotate(angle);

    line(0, 0, 40, 0);  

  }

23-09

210  Input 1: Mouse I

Using the mouseX and mouseY variables with an if structure allows the cursor to 
select regions of the screen. The following examples demonstrate the cursor making 
a selection between different areas of the display window.

  // Cursor position selects the left or right half

  // of the display window

  void setup() {

    size(100, 100);

    noStroke();

    fill(0);

  }

  void draw() {

    background(204);

    if (mouseX < 50) {       

      rect(0, 0, 50, 100);   // Left

    } else {                 

      rect(50, 0, 50, 100);  // Right 

    }

  }

  // Cursor position selects the left, middle, 

  // or right third of the display window

  void setup() {

    size(100, 100);

    noStroke();

    fill(0);

  }

  void draw() {

    background(204);

    if (mouseX < 33) {                             

      rect(0, 0, 33, 100);   // Left

    } else if ((mouseX >= 33) && (mouseX <= 66)) { 

      rect(33, 0, 33, 100);  // Middle

    } else {                                       

      rect(66, 0, 33, 100);  // Right

    }

  }

23-11

23-12

Reas_04_173-278.indd   Sec3:210Reas_04_173-278.indd   Sec3:210 5/16/07   10:05:13 AM5/16/07   10:05:13 AM

  // Cursor position selects a quadrant of 

  // the display window

  void setup() {

    size(100, 100);

    noStroke();

    fill(0);

  }

  void draw() {

    background(204);

    if ((mouseX <= 50) && (mouseY <= 50)) { 

      rect(0, 0, 50, 50);    // Upper-left

    } else if ((mouseX <= 50) && (mouseY > 50)) {   

      rect(0, 50, 50, 50);   // Lower-left

    } else if ((mouseX > 50) && (mouseY < 50)) {   

      rect(50, 0, 50, 50);   // Upper-right

    } else { 

      rect(50, 50, 50, 50);  // Lower-right

    }

  }

  // Cursor position selects a rectangular area to

  // change the fill color

  void setup() {

    size(100, 100);

    noStroke();

    fill(0);

  }

  void draw() {

    background(204);

    if ((mouseX > 40) && (mouseX < 80) && 

        (mouseY > 20) && (mouseY < 80)) {

      fill(255);

    } else {

      fill(0);

    }

    rect(40, 20, 40, 60);

  }

23-13

Mouse buttons

Computer mice and other similar input devices typically have between one and three 
buttons, and Processing can detect when these buttons are pressed. The button status 
and the cursor position together allow the mouse to perform different actions. For 
example, pressing a button when the mouse is over an icon can select it, so the icon can 
be moved to a different location on screen. The mousePressed variable is true if any 
mouse button is pressed and false if no mouse button is pressed. The variable 
mouseButton is LEFT, CENTER, or RIGHT depending on the mouse button most recently 
pressed. The mousePressed variable reverts to false as soon as the button is released, 
but the mouseButton variable retains its value until a different button is pressed. These 
variables can be used independently or in combination to control your software. Run 
these programs to see how the software responds to your fi ngers.

  // Set the square to white when a mouse button is pressed

  void setup() {

    size(100, 100);

  }

  void draw() {

    background(204);

    if (mousePressed == true) {

      fill(255);  // White   

    } else {

      fill(0);    // Black

    }

    rect(25, 25, 50, 50); 

  }

  // Set the square to black when the left mouse button 

  // is pressed and white when the right button is pressed

  void setup() {

    size(100, 100);

  }

  void draw() { 

    if (mouseButton == LEFT) { 

      fill(0);    // Black

    } else if (mouseButton == RIGHT) { 

      fill(255);  // White

    } else { 

23-15

213  Input 1: Mouse I

      fill(126);  // Gray

    } 

    rect(25, 25, 50, 50); 

  } 

  // Set the square to black when the left mouse button 

  // is pressed, white when the right button is pressed,  

  // and gray when a button is not pressed

  void setup() {

    size(100, 100);

  }

  void draw() { 

    if (mousePressed == true) {

      if (mouseButton == LEFT) { 

        fill(0);    // Black

      } else if (mouseButton == RIGHT) { 

        fill(255);  // White

      }

    } else { 

      fill(126);    // Gray

    }

    rect(25, 25, 50, 50);   

  }

Not all mice have multiple buttons, and if software is distributed widely, the interaction 
should not rely on detecting which button is pressed. For example, if you are posting 
your work on the Web, don’t rely on the middle or right button for using the software 
because many users won’t have a two- or three-button mouse.

Cursor icon

The cursor can be hidden with the noCursor() function and can be set to appear as a 
different icon with the cursor() function. When the noCursor() function is run, the 
cursor icon disappears as it moves into the display window. To give feedback about the 
location of the cursor within the software, a custom cursor can be drawn and controlled 
with the mouseX and mouseY variables. 

23-17

// Draw an ellipse to show the position of the hidden cursor

void setup() {

  size(100, 100);

  strokeWeight(7);

  smooth();

  noCursor();

}

void draw() {

  background(204);

  ellipse(mouseX, mouseY, 10, 10);

}

If noCursor() is run, the cursor will be hidden while the program is running until the 
cursor() function is run to reveal it.

// Hides the cursor until a mouse button is pressed

void setup() {

  size(100, 100);

  noCursor();

}

void draw() {

  background(204);

  if (mousePressed == true) {

    cursor();

  }

}

Adding a parameter to the cursor() function allows it to be changed to another icon. 
The self-descriptive options for the MODE parameter are ARROW, CROSS, HAND, MOVE, 
TEXT, and WAIT. 

// Draws the cursor as a hand when a mouse button is pressed

void setup() {

  size(100, 100);

  smooth();

}

void draw() {

  background(204);

23-18

215  Input 1: Mouse I

  if (mousePressed == true) {

    cursor(HAND);

  } else {

    cursor(MOVE);

  }

  line(mouseX, 0, mouseX, height);

  line(0, mouseY, height, mouseY);

}

These cursor images are part of your computer’s operating system and will appear 
differently on different machines. 

 Exercises
1.  Control the position of a shape with the mouse. Strive to create a more interesting  
 relation than one directly mimicking the position of the cursor.
2.  Invent three unique shapes that behave differently in relation to the mouse. Each  
 shape’s behavior should change when the mouse is pressed. Relate the form of 
 each shape to its behavior.
3.  Create a custom cursor that changes as the mouse moves through the 
 display window.

23-20
cont.

Extension 3:  Vision
Text by Golan Levin

A well-known anecdote relates how, sometime in 1966, the legendary artifi cial intelli-
gence pioneer Marvin Minsky directed an undergraduate student to solve “the problem 
of computer vision” as a summer project.1 This anecdote is often resuscitated to 
illustrate how egregiously the diffi culty of computational vision has been 
underestimated. Indeed, nearly forty years later the discipline continues to confront 
numerous unsolved (and perhaps unsolvable) challenges, particularly with respect to 
high-level “image understanding” issues such as pattern recognition and feature 
recognition. Nevertheless, the intervening decades of research have yielded a great 
wealth of well-understood, low-level techniques that are able, under controlled 
circumstances, to extract meaningful information from a camera scene. These 
techniques are indeed elementary enough to be implemented by novice programmers at 
the undergraduate or even high-school level.

Computer vision in interactive art 

The fi rst interactive artwork to incorporate computer vision was, interestingly enough, 
also one of the fi rst interactive artworks. Myron Krueger’s legendary Videoplace, 
developed between 1969 and 1975, was motivated by his deeply felt belief that the entire 
human body ought to have a role in our interactions with computers. In the Videoplace 
installation, a participant stands in front of a backlit wall and faces a video projection 
screen. The participant’s silhouette is then digitized and its posture, shape, and gestural 
movements analyzed. In response, Videoplace synthesizes graphics such as small 
“critters” that climb up the participant’s projected silhouette, or colored loops drawn 
between the participant’s fi ngers. Krueger also allowed participants to paint lines with 
their fi ngers, and, indeed, entire shapes with their bodies; eventually, Videoplace offered 
more than fi fty compositions and interactions. Videoplace is notable for many “fi rsts” in 
the history of human-computer interaction. Some of its interaction modules allowed two 
participants in mutually remote locations to participate in the same shared video space, 
connected across the network—an implementation of the fi rst multiperson virtual 
reality, or, as Krueger termed it, an “artifi cial reality.” Videoplace, it should be noted, was 
developed before the mouse became the ubiquitous desktop device it is today, and was 
(in part) created to demonstrate interface alternatives to the keyboard terminals that 
dominated computing so completely in the early 1970s. 
 Messa di Voce (p. 511), created by this text’s author in collaboration with Zachary 
Lieberman, uses whole-body vision-based interactions similar to Krueger’s, but combines 
them with speech analysis and situates them within a kind of projection-based Ra

548  Extension 3: Vision

augmented reality. In this audiovisual performance, the speech, shouts, and songs 
produced by two abstract vocalists are visualized and augmented in real time by 
synthetic graphics. To accomplish this, a computer uses a set of vision algorithms to 
track the locations of the performers’ heads; this computer also analyzes the audio 
signals coming from the performers’ microphones. In response, the system displays 
various kinds of visualizations on a projection screen located just behind the performers; 
these visualizations are synthesized in ways that are tightly coupled to the sounds being 
spoken and sung. With the help of the head-tracking system, moreover, these 
visualizations are projected such that they appear to emerge directly from the 
performers’ mouths.
  Rafael Lozano-Hemmer’s installation Standards and Double Standards (2004) 
incorporates full-body input in a less direct, more metaphorical context. This work 
consists of fi fty leather belts, suspended at waist height from robotic servomotors 
mounted on the ceiling of the exhibition room. Controlled by a computer vision-based 
tracking system, the belts rotate automatically to follow the public, turning their buckles 
slowly to face passers-by. Lozano-Hemmer’s piece “turns a condition of pure surveillance 
into an ‘absent crowd’ using a fetish of paternal authority: the belt.”2

 The theme of surveillance plays a foreground role in David Rokeby’s Sorting Daemon 
(2003). Motivated by the artist’s concerns about the increasing use of automated systems 
for profi ling people as part of the “war on terrorism,” this site-specifi c installation works 
toward the automatic construction of a diagnostic portrait of its social (and racial) 
environment. Rokeby writes: “The system looks out onto the street, panning, tilting and 
zooming, looking for moving things that might be people. When it fi nds what it thinks 
might be a person, it removes the person’s image from the background. The extracted 
person is then divided up according to areas of similar colour. The resulting swatches of 
colour are then organized [by hue, saturation and size] within the arbitrary context of 
the composite image” projected onsite at the installation’s host location.3

 Another project themed around issues of surveillance is Suicide Box, by the Bureau 
of Inverse Technology (Natalie Jeremijenko and Kate Rich). Presented as a device for 
measuring the hypothetical “despondency index” of a given locale, the Suicide Box 
nevertheless records very real data regarding suicide jumpers from the Golden Gate 
Bridge. According to the artists, “The Suicide Box is a motion-detection video system, 
positioned in range of the Golden Gate Bridge, San Francisco, in 1996. It watched the 
bridge constantly and when it recognized vertical motion, captured it to a video record. 
The resulting footage displays as a continuous stream the trickle of people who jump off 
the bridge. The Golden Gate Bridge is the premiere suicide destination in the United 
States; a 100-day initial deployment period of the Suicide Box recorded 17 suicides. 
During the same time period the Port Authority counted only 13.”4 Elsewhere, 
Jeremijenko has explained that “the idea was to track a tragic social phenomenon which 
was not being counted—that is, doesn’t count.”5 The Suicide Box has met with 
considerable controversy, ranging from ethical questions about recording the suicides to 
disbelief that the recordings could be real. Jeremijenko, whose aim is to address the 
hidden politics of technology, has pointed out that such attitudes express a recurrent 
theme—“the inherent suspicion of artists working with material evidence”—evidence 

obtained, in this case, with the help of machine vision-based surveillance.
 Considerably less macabre is Christian Möller’s clever Cheese installation (2003), 
which the artist developed in collaboration with the California Institute of Technology 
and the Machine Perception Laboratories of the University of California, San Diego. 
Motivated, perhaps, by the culture shock of his relocation to Hollywood, the German-
born Möller directed “six actresses to hold a smile for as long as they could, up to one and 
a half hours. Each ongoing smile is scrutinized by an emotion recognition system, and 
whenever the display of happiness fell below a certain threshold, an alarm alerted them 
to show more sincerity.”6 The installation replays recordings of the analyzed video on 
six fl at-panel monitors, with the addition of a fl uctuating graphic level-meter to indicate 
the strength of each actress’ smile. The technical implementation of this artwork’s 
vision-based emotion recognition system is quite sophisticated.
 As can be seen from these examples, artworks employing computer vision range 
from the highly formal and abstract to the humorous and sociopolitical. They concern 
themselves with the activities of willing participants, paid volunteers, or unaware 
strangers. They track people of interest at a wide variety of spatial scales, from extremely 
intimate studies of their facial expressions, to the gestures of their limbs, to the  
movements of entire bodies. The examples above represent just a small selection of 
notable works in the fi eld and of the ways in which people (and objects) have been 
tracked and dissected by video analysis. Other noteworthy artworks that use machine 
vision include Marie Sester’s Access; Joachim Sauter and Dirk Lüsebrink’s Zerseher and 
Bodymover; Scott Snibbe’s Boundary Functions and Screen Series; Camille Utterback and 
Romy Achituv’s TextRain; Jim Campbell’s Solstice; Christa Sommerer and Laurent 
Mignonneau’s A-Volve; Danny Rozin’s Wooden Mirror; Chico MacMurtrie’s Skeletal 
Refl ection, and various works by Simon Penny, Toshio Iwai, and numerous others. No 
doubt many more vision-based artworks remain to be created, especially as these 
techniques gradually become incorporated into developing fi elds like physical 
computing and robotics. 

Elementary computer vision techniques

To understand how novel forms of interactive media can take advantage of computer 
vision techniques, it is helpful to begin with an understanding of the kinds of problems 
that vision algorithms have been developed to address, and of their basic mechanisms of 
operation. The fundamental challenge presented by digital video is that it is 
computationally “opaque.” Unlike text, digital video data in its basic form—stored solely 
as a stream of rectangular pixel buffers—contains no intrinsic semantic or symbolic 
information. There is no widely agreed upon standard for representing the content of 
video, in a manner analogous to HTML, XML, or even ASCII for text (though some new 
initiatives, notably the MPEG-7 description language, may evolve into such a standard in 
the future). As a result, a computer, without additional programming, is unable to 
answer even the most elementary questions about whether a video stream contains a 
person or object, or whether an outdoor video scene shows daytime or nighttime, et 

cetera. The discipline of computer vision has developed to address this need.
 Many low-level computer vision algorithms are geared to the task of distinguishing 
which pixels, if any, belong to people or other objects of interest in the scene. Three 
elementary techniques for accomplishing this are frame differencing, which attempts to 
locate features by detecting their movements; background subtraction, which locates 
visitor pixels according to their difference from a known background scene; and 
brightness thresholding, which uses hoped-for differences in luminosity between 
foreground people and their background environment. These algorithms, described in 
the following examples, are extremely simple to implement and help constitute a base 
of detection schemes from which sophisticated interactive systems may be built. 

Example 1:  Detecting motion (p. 556)
The movements of people (or other objects) within the video frame can be detected and 
quantifi ed using a straightforward method called frame differencing. In this technique, 
each pixel in a video frame F1 is compared with its corresponding pixel in the 
subsequent frame F2. The difference in color and/or brightness between these two pixels 
is a measure of the amount of movement in that particular location. These differences 
can be summed across all of the pixels’ locations to provide a single measurement of the 
aggregate movement within the video frame. In some motion detection 
implementations, the video frame is spatially subdivided into a grid of cells, and the 
values derived from frame differencing are reported for each of the individual cells. For 
accuracy, the frame differencing algorithm depends on relatively stable environmental 
lighting, and on having a stationary camera (unless it is the motion of the camera that is 
being measured). 

Example 2:  Detecting presence (p. 557)
A technique called background subtraction makes it possible to detect the presence of 
people or other objects in a scene, and to distinguish the pixels that belong to them from 
those that do not. The technique operates by comparing each frame of video with a 
stored image of the scene’s background, captured at a point in time when the scene was 
known to be empty. For every pixel in the frame, the absolute difference is computed 
between its color and that of its corresponding pixel in the stored background image; 
areas that are very different from the background are likely to represent objects of 
interest. Background subtraction works well in heterogeneous environments, but it is 
very sensitive to changes in lighting conditions and depends on objects of interest 
having suffi cient contrast against the background scene. 

Example 3:  Detection through brightness thresholding (p. 559)
With the aid of controlled illumination (such as backlighting) and/or surface treatments 
(such as high-contrast paints), it is possible to ensure that objects are considerably 
darker or lighter than their surroundings. In such cases objects of interest can be 
distinguished based on their brightness alone. To do this, each video pixel’s brightness is 
compared to a threshold value and tagged accordingly as foreground or background. 

Example 1. Detects motion by comparing each video frame to the previous frame. The change is visualized 
and is calculated as a number.

Example 2. Detects the presence of someone or something in front of the camera by comparing each video 
frame with a previously saved frame. The change is visualized and is calculated as a number.

Example 3. Distinguishes the silhouette of people or objects in each video frame by comparing each pixel
to a threshold value. The circle is fi lled with white when it is within the silhouette.

Example 4. Tracks the brightest object in each video frame by calculating the brightest pixel. The light from 
the fl ashlight is the brightest element in the frame; therefore, the circle follows it. 

Example 4:  Brightness tracking (p. 560)
A rudimentary scheme for object tracking, ideal for tracking the location of a single 
illuminated point (such as a fl ashlight), fi nds the location of the single brightest pixel in 
every fresh frame of video. In this algorithm, the brightness of each pixel in the 
incoming video frame is compared with the brightest value yet encountered in that 
frame; if a pixel is brighter than the brightest value yet encountered, then the location 
and brightness of that pixel are stored. After all of the pixels have been examined, then 
the brightest location in the video frame is known. This technique relies on an 
operational assumption that there is only one such object of interest. With trivial 
modifi cations, it can equivalently locate and track the darkest pixel in the scene, or track 
multiple and differently colored objects.

Of course, many more software techniques exist, at every level of sophistication, for 
detecting, recognizing, and interacting with people and other objects of interest. Each of 
the tracking algorithms described above, for example, can be found in elaborated 
versions that amend its various limitations. Other easy-to-implement algorithms can 
compute specifi c features of a tracked object, such as its area, center of mass, angular 
orientation, compactness, edge pixels, and contour features such as corners and cavities. 
On the other hand, some of the most diffi cult to implement algorithms, representing the 
cutting edge of computer vision research today, are able (within limits) to recognize 
unique people, track the orientation of a person’s gaze, or correctly identify facial 
expressions. Pseudocodes, source codes, or ready-to-use implementations of all of these 
techniques can be found on the Internet in excellent resources like Daniel Huber’s 
Computer Vision Homepage, Robert Fisher’s HIPR (Hypermedia Image Processing 
Reference), or in the software toolkits discussed on pages 554-555.

Computer vision in the physical world

Unlike the human eye and brain, no computer vision algorithm is completely general, 
which is to say, able to perform its intended function given any possible video input. 
Instead, each software tracking or detection algorithm is critically dependent on certain 
unique assumptions about the real-world video scene it is expected to analyze. If any of 
these expectations are not met, then the algorithm can produce poor or ambiguous 
results or even fail altogether. For this reason, it is essential to design physical conditions 
in tandem with the development of computer vision code, and to select the software 
techniques that are most compatible with the available physical conditions. 
 Background subtraction and brightness thresholding, for example, can fail if the 
people in the scene are too close in color or brightness to their surroundings. For these 
algorithms to work well, it is greatly benefi cial to prepare physical circumstances that 
naturally emphasize the contrast between people and their environments. This can be 
achieved with lighting situations that silhouette the people, or through the use of 
specially colored costumes. The frame-differencing technique, likewise, fails to detect 
people if they are stationary. It will therefore have very different degrees of success 

detecting people in videos of offi ce waiting rooms compared with videos of the Tour de 
France bicycle race. 
 A wealth of other methods exist for optimizing physical conditions in order to 
enhance the robustness, accuracy, and effectiveness of computer vision software. Most 
are geared toward ensuring a high-contrast, low-noise input image. Under low-light 
conditions, for example, one of the most helpful such techniques is the use of infrared 
(IR) illumination. Infrared, which is invisible to the human eye, can supplement the light 
detected by conventional black-and-white security cameras. Using IR signifi cantly 
improves the signal-to-noise ratio of video captured in low-light circumstances and can 
even permit vision systems to operate in (apparently) complete darkness. Another 
physical optimization technique is the use of retrorefl ective marking materials, such as 
those manufactured by 3M Corporation for safety uniforms. These materials are 
remarkably effi cient at refl ecting light back toward their source of illumination and are 
ideal aids for ensuring high-contrast video of tracked objects. If a small light is placed 
coincident with the camera’s axis, objects with retrorefl ective markers will be detected 
with tremendous reliability.
 Finally, some of the most powerful physical optimizations for machine vision can be 
made without intervening in the observed environment at all, through well-informed 
selections of the imaging system’s camera, lens, and frame-grabber components. To take 
one example, the use of a “telecentric” lens can signifi cantly improve the performance of 
certain kinds of shape-based or size-based object recognition algorithms. For this type of 
lens, which has an effectively infi nite focal length, magnifi cation is nearly independent 
of object distance. As one manufacturer describes it, “an object moved from far away to 
near the lens goes into and out of sharp focus, but its image size is constant. This 
property is very important for gauging three-dimensional objects, or objects whose 
distance from the lens is not known precisely.”7 Likewise, polarizing fi lters offer a 
simple, nonintrusive solution to another common problem in video systems, namely 
glare from refl ective surfaces. And a wide range of video cameras are available, 
optimized for conditions like high-resolution capture, high-frame-rate capture, short 
exposure times, dim light, ultraviolet light, and thermal imaging. It pays to research 
imaging components carefully.
 As we have seen, computer vision algorithms can be selected to negotiate best the 
physical conditions presented by the world, and physical conditions can be modifi ed to 
be more easily legible to vision algorithms. But even the most sophisticated algorithms 
and the highest-quality hardware cannot help us fi nd meaning where there is none, or 
track an object that cannot be described in code. It is therefore worth emphasizing that 
some visual features contain more information about the world, and are also more 
easily detected by the computer, than others. In designing systems to “see for us,” we 
must not only become freshly awakened to the many things about the world that make 
it visually intelligible to us, but also develop a keen intuition about their ease of 
computability. The sun is the brightest point in the sky, and by its height also indicates 
the time of day. The mouth cavity is easily segmentable as a dark region, and the 
circularity of its shape is also closely linked to vowel sound. The pupils of the eyes emit 
an easy-to-track infrared retrorefl ection, and they also indicate a person’s direction of 

gaze. Simple frame differencing makes it easy to track motion in a video. The Suicide Box 
(p. 548) uses this technique to dramatic effect.

Tools for computer vision

It can be a rewarding experience to implement machine vision techniques from scratch 
using code such as the examples provided in this section. To make this possible, the only 
requirement of one’s software development environment is that it should provide direct 
read-access to the array of video pixels obtained by the computer’s frame-grabber. 
Hopefully, the example algorithms discussed earlier illustrate that creating low-level 
vision algorithms from fi rst principles isn’t so hard. Of course, a vast range of 
functionality can also be obtained immediately from readily available solutions. Some of 
the most popular machine vision toolkits take the form of plug-ins or extension libraries 
for commercial authoring environments geared toward the creation of interactive 
media. Such plug-ins simplify the developer’s problem of connecting the results of the 
vision-based analysis to the audio, visual, and textual affordances generally provided by 
such authoring systems.
 Many vision plug-ins have been developed for Max/MSP/Jitter, a visual 
programming environment that is widely used by electronic musicians and VJs. 
Originally developed at the Parisian IRCAM research center in the mid-1980s and now 
marketed commercially by the California-based Cycling’74 company, this extensible 
environment offers powerful control of (and connectivity between) MIDI devices, real-
time sound synthesis and analysis, OpenGL-based 3D graphics, video fi ltering, network 
communications, and serial control of hardware devices. The various computer vision 
plug-ins for Max/MSP/Jitter, such as David Rokeby’s SoftVNS, Eric Singer’s Cyclops, and 
Jean-Marc Pelletier’s CV.Jit, can be used to trigger any Max processes or control any 
system parameters. Pelletier’s toolkit, which is the most feature-rich of the three, is also 
the only one that is freeware. CV.Jit provides abstractions to assist users in tasks such as 
image segmentation, shape and gesture recognition, and motion tracking, as well as 
educational tools that outline the basics of computer vision techniques.
 Some computer vision toolkits take the form of stand-alone applications and are 
designed to communicate the results of their analyses to other environments (such as 
Processing, Director, or Max) through protocols like MIDI, serial RS-232, UDP, or TCP/IP 
networks. BigEye, developed by the STEIM (Studio for Electro-Instrumental Music) group 
in Holland, is a simple and inexpensive example. BigEye can track up to 16 objects of 
interest simultaneously, according to their brightness, color, and size. The software 
allows for a simple mode of operation in which the user can quickly link MIDI messages 
to many object parameters, such as position, speed, and size. Another example is the 
powerful EyesWeb open platform, a free system developed at the University of Genoa. 
Designed with a special focus on the analysis and processing of expressive gesture, 
EyesWeb includes a collection of modules for real-time motion tracking and extraction of 
movement cues from human full-body movement; a collection of modules for analysis of 
occupation of 2D space; and a collection of modules for extraction of features from 

trajectories in 2D space. EyesWeb’s extensive vision affordances make it highly 
recommended for students. 
 The most sophisticated toolkits for computer vision generally demand greater 
familiarity with digital signal processing, and they require developers to program in 
compiled languages like C++ rather than languages like Java, Lingo, or Max. The Intel 
Integrated Performance Primitives (IPP) library, for example, is among the most general 
commercial solutions available for computers with Intel-based CPUs. The OpenCV 
library, by contrast, is a free, open source toolkit with nearly similar capabilities and a 
tighter focus on commonplace computer vision tasks. The capabilities of these tools, as 
well as all of those mentioned above, are continually evolving.
 Processing includes a basic video library that handles getting pixel information 
from a camera or movie fi le as demonstrated in the examples included with this text. 
The computer vision capabilities of Processing are extended by libraries like Myron, 
which handles video input and has basic image processing capabilities. Other libraries 
connect Processing to EyesWeb and OpenCV. They can be found on the libraries page of 
the Processing website: www.processing.org/reference/libraries.

Conclusion 

Computer vision algorithms are increasingly used in interactive and other computer-
based artworks to track people’s activities. Techniques exist that can create real-time 
reports about people’s identities, locations, gestural movements, facial expressions, gait 
characteristics, gaze directions, and other attributes. Although the implementation of 
some vision algorithms requires advanced understanding of image processing and 
statistics, a number of widely used and highly effective techniques can be implemented 
by novice programmers in as little as an afternoon. For artists and designers who are 
familiar with popular multimedia authoring systems like Macromedia Director and 
Max/MSP/Jitter, a wide range of free and commercial toolkits are also available that 
provide ready access to more advanced vision functionalities. 
  Since the reliability of computer vision algorithms is limited according to the 
quality of the incoming video scene and the defi nition of a scene’s quality is determined 
by the specifi c algorithms that are used to analyze it, students approaching computer 
vision for the fi rst time are encouraged to apply as much effort to optimizing their 
physical scenario as they do their software code. In many cases, a cleverly designed 
physical environment can permit the tracking of phenomena that might otherwise 
require much more sophisticated software. As computers and video hardware become 
more available, and software-authoring tools continue to improve, we can expect to see 
the use of computer vision techniques increasingly incorporated into media-art 
education and into the creation of games, artworks, and many other applications.

 Notes

1.  http://mechanism.ucsd.edu/~bill/research/mercier/2ndlecture.pdf. 

2.  http://www.fundacion.telefonica.com/at/rlh/eproyecto.html. 

3.   http://homepage.mac.com/davidrokeby/sorting.html. 

4.   http://www.bureauit.org/sbox.

5.   http://www.wired.com/news/culture/0,1284,64720,00.html. 

6.   http://www.christian-moeller.com. 

7.  http://www.mellesgriot.com/pdf/pg11-19.pdf.

Code

Video can be captured into Processing from USB cameras, IEEE 1394 cameras, or video 
cards with composite or S-video input devices. The examples that follow assume you 
already have a camera working with Processing. Before trying these examples, fi rst get 
the examples included with the Processing software to work. Sometimes you can plug a 
camera into your computer and it will work immediately. Other times it’s a diffi cult 
process involving trial-and-error changes. It depends on the operating system, the 
camera, and how the computer is confi gured. For the most up-to-date information, refer 
to the Video reference on the Processing website: www.processing.org/reference/libraries.

Example 1:  Detecting motion

// Quantify the amount of movement in the video frame using frame-differencing

import processing.video.*;

int numPixels;

int[] previousFrame;

Capture video;

void setup(){

  size(640, 480);  // Change size to 320 x 240 if too slow at 640 x 480

  video = new Capture(this, width, height, 24);

  numPixels = video.width * video.height;

  // Create an array to store the previously captured frame   

  previousFrame = new int[numPixels];

}

void draw() {

  if (video.available()) {

    // When using video to manipulate the screen, use video.available() and 

    // video.read() inside the draw() method so that it's safe to draw to the screen

    video.read();        // Read the new frame from the camera

    video.loadPixels();  // Make its pixels[] array available

    int movementSum = 0; // Amount of movement in the frame    

    loadPixels();

    

    for (int i = 0; i < numPixels; i++) {  // For each pixel in the video frame...

      color currColor = video.pixels[i];         

      color prevColor = previousFrame[i];

      // Extract the red, green, and blue components from current pixel

      int currR = (currColor >> 16) & 0xFF;  // Like red(), but faster (see p. 673)         

      int currG = (currColor >> 8) & 0xFF;      

      int currB =  currColor & 0xFF;

      // Extract red, green, and blue components from previous pixel

      int prevR = (prevColor >> 16) & 0xFF;       

      int prevG = (prevColor >> 8) & 0xFF;      

      int prevB = prevColor & 0xFF;

      // Compute the difference of the red, green, and blue values

      int diffR = abs(currR - prevR);  

      int diffG = abs(currG - prevG);  

      int diffB = abs(currB - prevB);  

      // Add these differences to the running tally

      movementSum += diffR + diffG + diffB;

      // Render the difference image to the screen

      pixels[i] = color(diffR, diffG, diffB);     

      // The following line is much faster, but more confusing to read

      //pixels[i] = 0xff000000 | (diffR << 16) | (diffG << 8) | diffB;

      // Save the current color into the 'previous' buffer

      previousFrame[i] = currColor;    

    }

    

    // To prevent flicker from frames that are all black (no movement), 

    // only update the screen if the image has changed.

    if (movementSum > 0) {

      updatePixels();

      println(movementSum);  // Print the total amount of movement to the console

    }

  }

}

Example 2:  Detecting presence 

// Detect the presence of people and objects in the frame using a simple 

// background-subtraction technique. To initialize the background, press a key.

import processing.video.*;

int numPixels; 

int[] backgroundPixels;

Capture video;

void setup() {

  size(640, 480);  // Change size to 320 x 240 if too slow at 640 x 480

  video = new Capture(this, width, height, 24);

  numPixels = video.width * video.height;

  // Create array to store the background image

  backgroundPixels = new int[numPixels];

  // Make the pixels[] array available for direct manipulation

  loadPixels();

}

void draw() {

  if (video.available()) {

    video.read();        // Read a new video frame

    video.loadPixels();  // Make the pixels of video available

    

    // Difference between the current frame and the stored background

    int presenceSum = 0;

    for (int i = 0; i < numPixels; i++) {  // For each pixel in the video frame...      

      // Fetch the current color in that location, and also the color 

      // of the background in that spot

      color currColor = video.pixels[i];             

      color bkgdColor = backgroundPixels[i];    

      

   // Extract the red, green, and blue components of the current pixel’s color

      int currR = (currColor >> 16) & 0xFF;             

      int currG = (currColor >> 8) & 0xFF;           

      int currB = currColor & 0xFF;

      // Extract the red, green, and blue components of the background pixel’s color

      int bkgdR = (bkgdColor >> 16) & 0xFF;             

      int bkgdG = (bkgdColor >> 8) & 0xFF;             

      int bkgdB = bkgdColor & 0xFF;

      // Compute the difference of the red, green, and blue values

      int diffR = abs(currR - bkgdR);          

      int diffG = abs(currG - bkgdG);

      int diffB = abs(currB - bkgdB);

      // Add these differences to the running tally

      presenceSum += diffR + diffG + diffB;

      // Render the difference image to the screen

      pixels[i] = color(diffR, diffG, diffB);   

      // The following line does the same thing much faster, but is more technical

      //pixels[i] = 0xFF000000 | (diffR << 16) | (diffG << 8) | diffB;

    }

    updatePixels();        // Notify that the pixels[] array has changed

    println(presenceSum);  // Print out the total amount of movement

  }

}

// When a key is pressed, capture the background image into the backgroundPixels 

// buffer by copying each of the current frame’s pixels into it.

void keyPressed() {

  video.loadPixels();

  arraycopy(video.pixels, backgroundPixels);

}

Example 3:  Detection through brightness thresholding

// Determines whether a test location (such as the cursor) is contained within 

// the silhouette of a dark object

import processing.video.*;

color black = color(0);

color white = color(255);

int numPixels;

Capture video;

void setup() {

  size(640, 480);  // Change size to 320 x 240 if too slow at 640 x 480

  strokeWeight(5);

  video = new Capture(this, width, height, 24);

  numPixels = video.width * video.height;

  noCursor();

  smooth();

}

void draw() {

  if (video.available()) {

    video.read();

    video.loadPixels();

    int threshold = 127;    // Set the threshold value

    float pixelBrightness;  // Declare variable to store a pixel's color

    // Turn each pixel in the video frame black or white depending on its brightness

    loadPixels();

    for (int i = 0; i < numPixels; i++) {

      pixelBrightness = brightness(video.pixels[i]);

      if (pixelBrightness > threshold) {  // If the pixel is brighter than the 

        pixels[i] = white;                // threshold value, make it white

      } else {                            // Otherwise,

        pixels[i] = black;                // make it black

      }

    }

    updatePixels();

    // Test a location to see where it is contained. Fetch the pixel at the test 

    // location (the cursor), and compute its brightness

    int testValue = get(mouseX, mouseY);

    float testBrightness = brightness(testValue);

    if (testBrightness > threshold) {     // If the test location is brighter than

      fill(black);                        // the threshold set the fill to black

    } else {                              // Otherwise,

      fill(white);                        // set the fill to white

    }

    ellipse(mouseX, mouseY, 20, 20);   

  }

}

Example 4:  Brightness tracking

// Tracks the brightest pixel in a live video signal

import processing.video.*;

Capture video;

void setup(){

  size(640, 480);  // Change size to 320 x 240 if too slow at 640 x 480

  video = new Capture(this, width, height, 30);

  noStroke();

  smooth();

}

void draw() {

  if (video.available()) {

    video.read();

    image(video, 0, 0, width, height);  // Draw the webcam video onto the screen

    int brightestX = 0;        // X-coordinate of the brightest video pixel

    int brightestY = 0;        // Y-coordinate of the brightest video pixel

    float brightestValue = 0;  // Brightness of the brightest video pixel                

    // Search for the brightest pixel: For each row of pixels in the video image and     

    // for each pixel in the yth row, compute each pixel's index in the video

    video.loadPixels();

    int index = 0;

    for (int y = 0; y < video.height; y++) {

      for (int x = 0; x < video.width; x++) {

        // Get the color stored in the pixel 

        int pixelValue = video.pixels[index];

        // Determine the brightness of the pixel

        float pixelBrightness = brightness(pixelValue);  

        // If that value is brighter than any previous, then store the     

        // brightness of that pixel, as well as its (x,y) location

        if (pixelBrightness > brightestValue){           

          brightestValue = pixelBrightness;               

          brightestY = y;                    

          brightestX = x;

        }

        index++;

      }

    }

    // Draw a large, yellow circle at the brightest pixel

    fill(255, 204, 0, 128);

    ellipse(brightestX, brightestY, 200, 200);  

  }

}

Resources

Computer vision software toolkits
Camurri, Antonio, et al. Eyesweb. Vision-oriented software development environment. http://www.eyesweb.org.

Cycling’74 Inc. Max/MSP/Jitter. Graphic software development environment.  http://www.cycling74.com. 

Davies, Bob, et al. OpenCV. Open source computer vision library. http://sourceforge.net/projects/opencvlibrary. 

Nimoy, Joshua. Myron (WebCamXtra). Library (plug-in) for Macromedia Director and Processing.

 http://webcamxtra.sourceforge.net.

Pelletier, Jean-Marc. CV.Jit. Extension library for Max/MSP/Jitter. http://www.iamas.ac.jp/~jovan02/cv. 

Rokeby, David. SoftVNS. Extension library for Max/MSP/Jitter. 

 http://homepage.mac.com/davidrokeby/softVNS.html.

Singer, Eric. Cyclops. Extension library for Max/MSP/Jitter. http://www.cycling74.com/products/cyclops.html. 

STEIM (Studio for Electro-Instrumental Music). BigEye. Video analysis software. http://www.steim.org

Texts and artworks
Bureau of Inverse Technology. Suicide Box. http://www.bureauit.org/sbox.

Bechtel, William. The Cardinal Mercier Lectures at the Catholic University of Louvain. Lecture 2, An Exemplar. 

Neural Mechanism: The Brain’s Visual Processing System. 2003, p.1. 

 http://mechanism.ucsd.edu/~bill/research/mercier/2ndlecture.pdf. 

Fisher, Robert, et. al. HIPR (The Hypermedia Image Processing Reference).

 http://homepages.inf.ed.ac.uk/rbf/HIPR2/index.htm. 

Fisher, Robert, et al. CVonline: The Evolving, Distributed, Non-Proprietary, On-Line Compendium of 

 Computer Vision. http://homepages.inf.ed.ac.uk/rbf/CVonline. 

Huber, Daniel, et al. The Computer Vision Homepage. http://www-2.cs.cmu.edu/~cil/vision.html. 

Krueger, Myron. Artifi cial Reality II. Addison-Wesley Professional, 1991.

Levin, Golan and Lieberman, Zachary. Messa di Voce. Interactive installation, 2003.

 http://www.tmema.org/messa.

Levin, Golan, and Zachary Lieberman. “In-Situ Speech Visualization in Real-Time Interactive Installation   

 and Performance.” Proceedings of the Third International Symposium on Non-Photorealistic Animation and   

 Rendering. Annecy, France, June 7-9, 2004. http://www.fl ong.com/writings/pdf/messa_NPAR_2004_150dpi.pdf. 

Lozano-Hemmer, Rafael. Standards and Double Standards. Interactive installation.

 http://www.fundacion.telefonica.com/at/rlh/eproyecto.html. 

Melles Griot Corporation. Machine Vision Lens Fundamentals. http://www.mellesgriot.com/pdf/pg11-19.pdf.

Möller, Christian. Cheese. Installation artwork, 2003. http://www.christian-moeller.com. 

Rokeby, David. Sorting Daemon. Computer-based installation, 2003.

 http://homepage.mac.com/davidrokeby/sorting.html. 

Shachtman, Noah. “Tech and Art Mix at RNC Protest.” Wired News, 27 August 2004.

 http://www.wired.com/news/culture/0,1284,64720,00.html. 

Sparacino, Flavia. “(Some) computer vision based interfaces for interactive art and entertainment installations.”   

 INTER_FACE Body Boundaries, issue edited by Emanuele Quinz. Anomalie no. 2. Anomos, 2001.

 http://www.sensingplaces.com/papers/Flavia_isea2000.pdf. 

703  

Code Index 

This index contains all of 
the Processing language 
elements introduced 
within this book. The 
page numbers refer to 
the fi rst use.

! (logical NOT), 57
!= (inequality), 52
% (modulo), 45
&& (logical AND), 57
() (parentheses)
  for functions, 18
  for precedence, 47
* (multiply), 44
*= (multiply assign), 49
+ (addition), 43
++ (increment), 48
+= (add assign), 48
, (comma), 18
- (minus), 44
-- (decrement), 48
-= (subtract assign), 48
. (dot), 107
/ (divide), 44
/= (divide assign), 49
/* */ (comment), 18
// (comment), 17
; (semicolon), 19
< (less than), 51
<= (less than or 
    equal to), 52
= (assign), 38
== (equality), 52
  for String objects, 109
> (greater than), 51
>= (greater than 
    or equal to), 52
[] (array access), 301
  2D arrays, 312
  arrays of objects, 406
{} (braces), 53
  and variable scope, 178
|| (logical OR), 57
# (hex color), 93

abs(), 241
alpha(), 338
ambient(), 533
ambientLight(), 533
append(), 309
arc(), 124
arraycopy, 310
Array, 301
  length, 304
atan2(), 243

background(), 31
beginRaw(), 531
beginRecord(), 607
beginShape(), 69
bezier(), 30
bezierVertex(), 75
blend(), 351
blendColor(), 352
blue(), 337
boolean, 38
boolean(), 106
brightness(), 338
byte, 38
byte(), 106

camera(), 531
Capture, 556
ceil(), 49
char, 38, 102
char(), 106
class, 395
Client, 567
color, 38, 89
color(), 89
colorMode(), 91
constrain(), 237
copy(), 353
cos(), 118
createGraphics(), 614
createImage(), 362
createWriter(), 423
cursor(), 213
curveVertex(), 74

day(), 249
degrees(), 117
directionalLight(), 536
dist(), 238
draw(), 173

ellipse(), 30
ellipseMode(), 34
else, 55
else if, 56
endRaw(), 531
endRecord(), 607
endShape(), 69
exit(), 422
expand(), 309
extends, 456

false, 38
fill(), 32
filter(), 347
float, 37
float(), 106
floor(), 49
for, 61
frameCount, 173
frameRate(), 173

get(), 321
green(), 337

HALF_PI, 117
height, 40
hour(), 245
HSB, 89
hue(), 338

if, 53
image(), 96
int, 37
int(), 107
key, 225
keyCode, 227
keyPressed, 224
keyPressed(), 232
keyReleased(), 232

lerp(), 81
lightSpecular(), 536
line(), 27
loadFont(), 112
loadImage(), 96
loadPixels(), 356
loadStrings(), 428
loop(), 235

map(), 81
mask(), 354
max(), 50
millis(), 248
min(), 50
minute(), 245
month(), 249
mouseButton, 212
mouseDragged(), 229
mouseMoved(), 229
mousePressed, 212
mousePressed(), 229
mouseReleased(), 229
mouseX, 205
mouseY, 205

new
  for arrays, 303
  for objects, 399
nf(), 422
noCursor(), 213
noFill(), 33
noise(), 130
noiseSeed(), 131
noLoop(), 178
norm(), 80
noSmooth(), 33
noStroke(), 33
noTint(), 97

Object, 107, 395

PFont, 112
PI, 117
PImage, 96
pixels[], 356
pmouseX, 208
pmouseY, 208
point(), 25
pointLight(), 536
popMatrix(), 134
pow(), 80
print(), 20
println(), 20
PrintWriter, 423
  close(), 423
  flush(), 423
  println(), 424
pushMatrix(), 134

quad(), 29
QUARTER_PI, 117

radians(), 117
random(), 127
randomSeed(), 129
rect(), 29
rectMode(), 34
red(), 337
redraw(), 235
return, 194
RGB, 89
rotate(), 137
round(), 50

saturation(), 338
save(), 368
saveFrame(), 369
saveStrings(), 422
scale(), 138
second(), 245
Server, 567
set(), 324
setup(), 177
shorten(), 309
sin(), 118
size(), 24
  with P3D, 528   
 with OPENGL, 528
  with PDF, 607
smooth(), 33
specular(), 536
split(), 429
splitTokens(), 430
spotLight(), 536
sq(), 79
sqrt(), 79
str(), 107
String, 103
  length(), 108
  endsWith(), 108
  equals(), 109
  startsWith(), 108
  substring(), 109
  toCharArray(), 108
  toLowerCase(), 109
  toUpperCase(), 109
stroke(), 32
strokeCap(), 33
strokeJoin(), 33
strokeWeight(), 33
super, 456

text(), 112
textAlign(), 115
textFont(), 112
textLeading(), 115
textSize(), 114
texture(), 536
textWidth(), 116
tint(), 97
translate(), 133
triangle(), 27
true, 38
TWO_PI, 117

updatePixels(), 356

vertex(), 69
void, 187

width, 40

year(), 249

705  

Index 

This index contains mostly 
people,  software, artwork, 
and programming 
languages. For topics, 
see the table of contents 
(pp. vii–xvii); for code, see 
the Code Index.

1:1 (Jevbratt), 566 
3M Corporation, 553
3 Stoppages Étalon (Duchamp), 127
7–11 Email list, 563

AARON, 218
Aesthetics and Computation   
 Group (ACG), xxiii, 682
Achituv, Romy, 549
ActionScript, 158, 166, 522–523, 565,  
 680–681, 686–687, 689, 691
Adair, Sandra, 384
Adobe, 4, 169, 683
Adobe After Effects, 166, 327, 
 379, 387
Adobe Flash, 157–158, 165–166,   
 267–268, 275, 278, 327, 436, 
 564–565, 624, 629, 642, 
 680–681, 683, 686, 701
Adobe Flash Lite, 624, 681
Adobe Garamond (font), 112
Adobe Illustrator, xxiii, 30, 77, 166,   
 143, 217, 271, 273, 607–608, 683
Adobe Photoshop, xxiii, 95, 166,
  268, 276, 347, 355, 360, 384,   
  387–388, 391–392, 607–608,   
  611, 683
Adobe Premiere, 391–392
Adobe Streamline, 166
AAC (Advanced Audio Coding),   
 585 
AIFF (Audio Interchange File   
 Format), 585–586, 699
Aldus PageMaker, 605
Alexander, Ryan, 380
Alias Maya, 379, 387–388, 537, 680
AltSys, 170
Andrade, Laura Hernandez, 4
Apple IIe, xxiii
Apple Audio Units (AU), 591

Apple Computer, 3, 111, 537, 
 585, 699
Apple Logic Audio, 503, 591
Apple Mac G3, 383
Apple Mac G4, 383
Apple Macintosh (Mac), 9–11, 95,   
 111–112, 169, 205, 227, 367, 383, 
 521, 568–569, 574, 604, 639, 665,   
 682, 685
Apple Mac Mini, 639
Apple Mac OS, 264, 435, 665–666,   
 681 
Apple Mac OS X, 16, 170, 435, 645,   
 649, 684
Apple QuickTime, 367, 383–384,   
 387–388
AppleScript, 681
Arduino, 521, 633, 640, 641, 645–646,  
 648–649, 681, 685
Arp, Jean, 127
Ars Electronica Festival, 618
ART+COM, 498
ASCII (American Standard Code for 
Information Interchange), 102–103,  
 226–227, 549, 565, 664–668, 670,   
 691, 699
Athena, 387
ATI, 537
AT&T/Bell , 564
Audacity, 591
AutoCAD, 217, 529, 537
Autodesk 3ds Max, 268, 276,   
 391–392, 537
AutoDesk Revit, 537
AutoLISP, 522, 681
Autonomedia, 564 
Avid/Digidesign Pro Tools, 591
AVR (Atmel), 640
awk, 517, 684

Babbitt, Milton, 580–581
Bach, J. S., 581
Bailey, Chris, 581
Balkin, Amy, 267
Baran, Paul, 564
Barr, Alfred, 291
Barragán, Hernando, 633 
BASIC, xxiii, xxiv, 152, 264, 522, 
 604–605, 640, 642, 681
BASIC Stamp 2 (Parallax), 640
BasicX–24 (NetMedia), 642 

Bass, Saul, 327
Baumgärtel, Tilman, 564
Bauhaus, 149
BBC Acorn Archimedes,  264
Beach Culture, 605
Beethoven, Ludwig van, 581
BEFLIX, 315, 681
Bell Laboratories, 315, 580–581, 604
Bentley Systems     
 GenerativeComponents, 537
Berliner, Emile, 579
Berlow, David, 170
Bernard (a k a Flip 1), 508
BIAS Peak, 591
BigEye, 554
Binary Runtime Environment for   
 Wireless (BREW), 625
Binary space partition (BSP), 527
Binder, Maurice, 327
bitforms gallery, 164, 166–167, 525,   
 547, 603, 633
Bittorent, 571
Blackwell, Lewis, 605 
Blender, 276, 576
Blinkenlights (Chaos Computer   
 Club), 618
Blonk, Jaap, 511
Bluetooth, 619, 621–622, 624, 641,   
 645, 683
Blyth, Steven, 512
Boids (Reynolds), 295, 473,   
 475, 497 
Boole, George, 38, 61, 669
Boolean algebra, 38
Boulez, Pierre, 581 
Braitenberg, Valentino, 473–474
Brakhage, Stan, 413
Brecht, Bertolt, 564
Brooklyn Academy of Music   
 (BAM), 515–516
Brown, Robert, 295
Brownian motion, 295
Brunelleschi, Filippo, 525
Bunting, Heath, 563–564 
Bureau of Inverse Technology,   
 548, 634
Burke, Phil, 592
Burton, Ed, 263–264, 413, 499
Byrne, David, 581

C, 7, 264, 515–517, 522–523, 592, 640,
 642, 682–685, 693, 697
C++,  264, 271, 383, 507–508, 511–512,
 515–516, 522–523, 555, 592, 599,   
 640, 679, 681–682
CAD (computer–aided drawing   
 software), 217, 526, 537–538
Cage, John, 127, 579
CalArts School of Art, 564
California Institute of Technology   
 (Caltech), 388, 549
Cameron, Dan, 387
Campbell, Jim, 549
Carmack, John ,525
Carnegie Mellon University, xxi
Carnivore, 566, 568–569
Carson, David, 605
Cascading Style Sheets (CSS), 93 
CCRMA Synthesis ToolKit (STK),   
 592
Chang, Zai, 6
Cheese (Möller), 549
Cho, Peter, 257, 327
CIA World Fact Book, 267
Citron, Jack, 315
CityPoems, 617, 624
ChucK, 592, 682
Cloaca (Delvoye), 461
Clash of the Titans, 387
Close, Chuck, 606
CODE (Petzold), 648
Cohen, Harold, 218
Columbia–Princeton Electronic   
 Music Center, 580
Commodore C–64, 272
Commodore VC–20, 272
Common Lisp, 592
Complexification.net, 6, 157
Computational Beauty of Nature,   
 The (Flake), 469
Computers and Automation, 603
Computer Clubhouse, 680
Computer Lib / Dream Machines   
 (Nelson), 3
Computer Vision Homepage   
 (Huber), 552
Coniglio, Mark, 512
“Constituents for a Theory of the   
 Media” (Enzensberger), 564
Conway, John, 461, 463, 467–468,   
 475
Cook, Perry, 592  
Cooper, Muriel , 327
Cope, David, 581
CorelDRAW, 608

Cosic, Vic 563–564
Costabile, Sue (SUE.C), 503–504
Craighead, Alison, 618
Crawford, David, 316
Crystal Castle, 525
Csikszentmihályi, Chris, 507–508,   
 634
CSIRAC, 580
Csuri, Charles, 217
Cuba, Larry, 1, 315
Cullen, Mathew, 379–380
CV.Jit, 554
Cybernetic Serendipity, 101, 603
Cycling ’74, 554, 592
Cyclops, 554 

Dada, 149–150
Davies, Char, 526
Davis, Joshua, 564–565 
Deck, Barry, 112
Deleuze and Guattari, 564
Delvoye, Wim, 461
De Mol, Gerry, 275
Design By Numbers (DBN), xxiv,   
 552–523, 682
Designers Republic, The, 605
Dextro, 316
Dialtones (Levin et al.), 617–618
Digidesign, 587, 591
Dine, Jim, 606
DJ I, Robot Sound System, 506–509
Dodgeball, 617, 624
Domain Name System (DNS), 566
DrawBot, 169, 682, 684
Drawing with Computers (Wilson),  
 152, 217, 604
Drawn (Lieberman), 413
DuBois, R. Luke, 579
Duchamp, Marcel, 127, 633
Dunne, Tony, 634
Dürer, Albrecht 525, 612
DXF, 520, 529–531
Dynabook, 3

Eagle, 272
écal (école cantonale d’art de 
Lausanne), 271
Eclipse, 571, 625
ECMAScript, 681, 683
Edelweiss Series (Maywa Denki),   
 634
Edgerton, Harold, 295
Edison, Thomas, 579 
Eighth Istanbul Biennial, 387
Eimart, Herbert, 580

Electronic Arts, 585
ELIZA, 101
Emacs, 516
Emigre, 605
End of Print, The (Blackwell), 605
Endless Forest, The (Tale of Tales),   
 274–277
Engelbart, Douglas, 205
Eno, Brian, 581
Enron, 268
Enzensberger, Hans Magnus, 564
EPS, 606
Euler’s method, 7, 494
Every Icon (Simon), 565
Evolved Virtual Creatures (Sims),   
 295
Experiments in Art and   
 Technology (E.A.T.), 633
Extend Script, 683
Eye magazine, 605
Eye Catching (Steinkamp), 
 386–389 
EyesWeb, 554–555
EZIO (NIQ), 642 

Feingold, Ken, 633
Ferro, Pablo, 327
Final Cut Pro (FCP), 383, 503
Final Scratch, 507
Fischinger, Oskar, 413
Fisher, Robert, 552 
Flake, Gary William, 469
Flight404.com, 6
Flight Simulator, 525
Foldes, Peter, 315
FontLab, 170
Fontographer, 170
Fortran, 522
Fractal.Invaders (Tarbell), 156–159 
Franceschini, Amy, 267
Franke, Uli, 260, 271
Free Radicals, 413
Friendster, 617
Fourier, Jean-Baptiste-Joseph, 584
Fourier transform, 585, 588, 590
Futurist, 279, 579

Gabo, Nam, 633
Galloway, Alexander R., 563
Game of Life, 461, 463, 465–466,   
 468, 475
Gardner, Martin, 461, 463
Garton, Brad, 581 
Gerhardt, Joseph, 391–392
Gestalt psychology, 584

GIF, 95–96, 98–99, 421, 700–701
Girroir, Jonathan, 506–509
Google, 568, 617
GPS (Global positioning system),
 619, 621
Graffiti, 223
GRASS, 681
Groeneveld, Dirk, 333
GNU Image Manipulation   
 Program (GIMP), 95, 347, 355, 
 607–608 
GNU Public License (GPL), 271
Gnutella, 566, 571
GPU (graphics processing unit),    
 536–537
Graphomat Z64 (Zuse), 603 
Greenwold, Simon, 525
Greie, Antye (AGF), 503–504
Grzinic, Marina, 563
GUI (Graphical user interface),   
 435–436, 448, 450, 499, 604,   
 634, 679–680, 683, 685, 700
Gutenberg, Johannes, 111
Gutenberg archive, 433
Guttmann, Newmann, 580
Gysin, Andreas, 373

Hall, Grady, 379
Handel, George Frideric, 581
Hansen, Mark, 515–516, 634
Harmon, Leon, 604
Harvard University, xxi
Harvey, Auriea, 275
Hewlett-Packard (HP), 604, 610
Hawkinson, Tim, 633
Hawtin, Richie, 507
Hébert, Jean-Pierre, 217, 606
Hektor (Lehni, Franke), 260,   
 270–273
Henry, John, 507
Henry, Pierre, 580
Hiller, Lejaren , 581
Hoefler, Jonathan, 112
Hodgin, Robert, 6, 692
Hokusai, 612
Hongik University, 5
Hong, Leon, 5, 375
Hooke’s law, 263, 487
Howard Wise gallery, 603
HTML (HyperText Markup 
Language), 9–11, 93, 268, 427, 549,   
 564–565, 568–569, 621, 624,   
 665–666, 684
HTTP (Hypertext Transfer   
 Protocol), 567–569, 623

Huber, Daniel, 552
Huff, Kenneth A., 606
Hypermedia Image Processing 
Reference (HIPR), 552
HyperTalk, 522

IANA, 569
IBM, 315, 537, 580, 585, 604, 
 620, 702
IC (integrated circuit), 639, 647
I–Cube X (Infusion Systems), 642
IEEE 1394 camera, 556
If/Then (Feingold), 633
Igarashi, Takeo, 538
Igoe, Tom, 635, 648  
Ikarus M, 170
Incredibles, The, 315
Internet Explorer, 565
Internet Protocol (IP), 566–567,   
 569, 589, 645 
Impressionist, 279
Inaudible Cities: Part One   
 (Semiconductor), 392
InDesign, 683
Infrared, 553, 621
Inge, Leif, 581
Inkscape, 77, 607–608
Installation (Greenwold), 526
Institute of Contemporary Arts   
 (ICA), 101, 522
Intel Integrated Performance   
 Primitives (IPP), 512, 555
Interaction Design Institute Ivrea   
 (IDII), xxi, 634
i|o 360°, 565
I/O/D 4 (“The Webstalker”), 566
IRCAM, 554, 581, 592
Ishii, Hiroshi, 634
Ishizaki, Suguru, 327
ISO 216 standard, 611
Iwai, Toshio, 512, 549

James, Richard (Aphex Twin), 582 
Jarman, Ruth, 391–392
Java, 7, 9–11, 146, 161–162, 263–264,   
 271, 499, 521–523, 528, 555, 564–  
 565, 571, 574, 592, 622, 625– 626,   
 642, 663, 673, 677, 679–683,   
 686–690, 699–700
Java 2 Micro Edition (J2ME), 625
Java applet, 9–11, 264, 521, 656, 657,  
 675, 699
Java Archive (JAR), 10–11, 700
Java Core API, 271

JavaScript, 268, 271, 522, 624, 680,   
 681, 683
Java Virtual Machine (JVM), 680
Jeremijenko, Natalie,  548
Jevbratt, Lisa, 566 
jMax, 592
Jodi, 563–566    
Jones, Crispin, 634
Jones, Ronald, 275
Jonzun Crew, 508
JPEG, 95–96, 162, 421, 606, 611,   
 620, 701
JSyn (Java Synthesis), 592
Julesz, Bela, 603  

Kay, Alan, 3
Kim, Tai-kyung, 5
Kimura, Mari, 582
King’s Quest, 525
Klee, Paul, 217
Knowlton, Kenneth C., 315, 604
Krueger, Myron, 255, 512, 547
Kusaite, Lina, 275
Kuwakubo, Ryota, 634

La Barbara, Joan, 511
Langton, Chris, 469, 471
Putto8 2.2.2.2 (Rees), 524, 526
LaserWriter, 111, 604
Lee, Soo-jeong, 5
Led Zeppelin, 161
Legible City, The (Shaw,   
 Groeneveld), 333
Lehni, Jürg, 260, 271–273
Leibniz, Gottfried Wilhelm, 61
Letterscapes (Cho), 327
LettError, 111, 168–170, 605
Levin, Golan, 259, 333, 511–512, 
 547, 617–618
Lewis, George, 582
LeWitt, Sol, 217
Li, Francis, 617
Lia, 316, 496
Lialina, Olia, 563–564
Licko, Zuzana, 112, 605
Lieberman, Zachary, 413, 
 512–512, 547
Lifestreams, 425–426
Limewire, 571
Lingo, 522–523, 555, 565, 683, 
 686–687, 689, 691
Linklater, Richard, 383
Linotype, 111
Linux, 4, 9–11, 508, 521, 568–569,   
 625, 645, 649

Listening Post (Rubin, Hansen),   
 514–517
LISP, 101
LiveScript, 683
Local area network (LAN), 
 568–569
Logo, xxiii, 2, 217, 522, 681
Lovink, Geert, 564
Lozano-Hemmer, Rafael, 546, 548 
Lucent Technologies, 515
Lucier, Alvin, 590
Luening, Otto, 580
Lüsebrink, Dirk, 549
Lye, Len, 413

Machine Art exhibition, 291, 633
Machine Perception Laboratories,   
 549
MacMurtrie, Chico, 549 
Macromedia Director, 166, 
 387–388, 554–555, 642, 683, 686
Maeda, John, xix, xxiii, xxiv, 3, 5,   
 158, 333, 564, 606, 682
Malka, Ariel, 372
Makela, P. Scott, 605
Mandelbrot, Benoit, 153
Manovich, Lev, 565
Marble Madness, 525
Marconi, Guglielmo, 579
Marey, Étienne-Jules, 295
Mark of the Unicorn Digital   
 Performer, 591
Markov chain, 581
Marx, Karl, 267–268
Massachusetts Institute of   
 Technology (MIT), xix, xxiii,   
 xxiv, 327, 634, 680, 682, 693, 695
Masterman, Margaret, 101
Mathews, Max, 580, 586, 591, 683
MATLAB, 522
Max/MSP/Jitter, 2, 503–504, 
 515–517, 522, 554–555, 571, 580,   
 592, 642, 683–685
Maya Embedded Language (MEL),   
 680, 683
Maywa Denki, 634
McCarthy, John, 101
McCartney, James, 592
McCay, Winsor, 315
McLaren, Norman, 413
Medusa, 387
MEL, 680, 683
Mendel, Lucy, 507
Messa di Voce (Tmema et al.), 
 510–513, 547

Metrowerks Codewarrior, 512
Microsoft, 4, 111, 169, 436, 508, 
 525, 537, 585, 702
Microsoft Direct3D, 537
Microsoft Visual Basic, 436
Microsoft Windows, 9, 11, 264, 367,   
 421, 435–436, 511, 521, 568, 625,   
 645, 649, 665–666, 685
MIDI (Musical Instrument Digital   
 Interface) 162, 554, 588–589,   
 591–592, 618, 621, 623, 642, 645, 
 683, 685

